Resolución de Problemas y Algoritmos

Clase 3
Programación en Pascal.
Tipos de datos. Expresiones.



George Boole

Dr. Alejandro J. García
http:// cs.uns.edu.ar /~ajg

Departamento de Ciencias e Ingeniería de la Computación
Universidad Nacional del Sur
Bahía Blanca - Argentina

Conceptos de las clases anteriores

1. **Algoritmo.**
   Primitiva.
   Traza.
2. **Lenguaje de programación.**
   Pascal:
   - **Identificadores**
   - **Constantes y variables**
   - **Tipos de datos.**
   - **Primitiva de asignación (:=)**
   - **Primitivas read y write**



Una pregunta: ¿Cómo escribo el signo que abre en la consola?

En esta clase veremos herramientas para eso.

Herramientas a disposición de los alumnos

En RPA ofrecemos a los alumnos estas herramientas que se complementan unas a otras:

1. **clase teórica** (presentación grupal, discusión grupal, reflexión, análisis, propuesta de metodologías)
2. **ejercicios en los prácticos** (trabajo individual, puesta en práctica de conceptos y metodologías)
3. **clase práctica** (atención personalizada para discusión y comprobación de los resultados obtenidos, puesta en común, reflexión, consultas ejecutando en computadoras)
4. **material en línea** (http://cs.uns.edu.ar/~wmg/rpalz/)
5. **evaluación** en máquina y parciales. Pone una meta en una fecha fija que ayuda a organizarse y poner en práctica las habilidades desarrolladas, como en la vida profesional. Además, el alumno recibe una devolución formal escrita sobre su desempeño.

Objetivos de la materia RPA

El objetivo principal de RPA es que los alumnos adquieran la capacidad de desarrollar programas de computadoras para resolver problemas de pequeña escala.

El desarrollo de un programa se concibe como un proceso que abarca varias etapas:

1. La interpretación adecuada del enunciado a través del cual se plantea el problema.
2. El diseño de un algoritmo que especifica la resolución del problema.
3. La implementación del algoritmo en un lenguaje de programación imperativo.
4. La verificación de la solución.

Etapas

El objetivo principal de RPA es que los alumnos adquieran la capacidad de desarrollar programas de computadoras para resolver problemas de pequeña escala

El desarrollo de un programa se concibe como un proceso que abarca varias etapas:

1. La interpretación adecuada del enunciado a través del cual se plantea el problema.
2. El diseño de un algoritmo que especifica la resolución del problema.
3. La implementación del algoritmo en un lenguaje de programación imperativo.
4. La verificación de la solución.

Estas etapas están en sintonía con el proceso de ingeniería de software que veremos más adelante

Concepto: lenguaje de programación

Un **lenguaje de programación** es un lenguaje artificial creado para expresar procesos que pueden ser llevados a cabo por computadoras. (Ejemplo: Pascal)

Un lenguaje de programación está definido por:

1. **un conjunto de símbolos**,
2. **reglas sintácticas** que definen su estructura, y
3. **reglas semánticas** que definen el significado de sus elementos.

Conceptos: Diagrama Sintáctico

- La **sintaxis** de un lenguaje de programación es un conjunto de reglas que indica la estructura de los programas en ese lenguaje.

Un **diagrama sintáctico** (syntax diagram or railroad diagrams) es una forma gráfica de representar la sintaxis de un lenguaje de programación. Permite **describir sin ambigüedad** la sintaxis de un lenguaje de una manera simple y formal.

- Los diagramas sintácticos de Pascal que usaremos en RPA se encuentran en la página de la materia siguiendo este enlace: http://cs.uns.edu.ar/~wmg/rpalz/downloads/Practicos/Diagramas Sintacticos Pascal.pdf
- La sintaxis original escrita por N. Wirth está en la pág. 47 de : http://e-collection.library.ethz.ch/eserv/eth:3059/eth-3059-01.pdf

Elementos de un diagrama sintáctico

nombre → (1) **Un nombre y flecha** indican el comienzo de un diagrama para la definición de nombre.

texto — (2) **Las figuras "redondeadas"** indican que texto se debe incluir tal cual como aparece.

nombre — (3) Los **rectángulos** indican que nombre está definido en algún otro diagrama sintáctico.

→ (4) Las **flechas** indican el orden de lectura en el diagrama.

Todos los programas en Pascal tienen esta estructura sintáctica:



Diagramas sintácticos

Todos los programas en Pascal tienen esta estructura sintáctica:



Ejemplo:

```
PROGRAM AreaCirculo;
CONST pi = 3.1416;
VAR area,radio: REAL;
BEGIN
 write('Ingrese radio:'); read(radio);
 area := pi * radio * radio;
 write('Area es', area);
END.
```



Parte del archivo disponible en la página de RPA



Diagramas sintácticos para "identificador"



- En letra se puede utilizar el símbolo "_" (underscore), las mayúsculas: ABCDEFGHIJKLMNOPQRSTUVWXYZ y las minúsculas: abcdefghijklmnopqrstuvwxyz
- No pueden utilizarse por ejemplo: á, ó, Ú, ñ, Ñ, -.

Metodología general propuesta



Observación importante: los diagramas sintácticos nos ayudan a entender la sintaxis del lenguaje y poder identificar *errores sintácticos*.

Un programa puede ser sintácticamente válido (cumple con todas las reglas de sintaxis) pero igualmente tener errores que no son sintácticos. Para poder identificar errores en la lógica del programa se puede utilizar una traza.

Primitiva de asignación

En una asignación: variable := expresión
1) primero se **evalúa** la expresión de derecha y se obtiene un valor,
2) luego se **modifica el valor** de la variable, **perdiéndose el valor anterior.**

```
PROGRAM Asigna2;
CONST c = 2.3;
VAR a, b: REAL;
BEGIN
  b:= 10;
  a:= 1;
  a:= a + 1;
  a:= a + 1;
  a:= a + 1;
END.
```

Traza de los valores almacenados en memoria para cada variable:

a	b
?	?
?	10
1	10
2	10
3	10
4	10

"?" indica "sin valor"

Primitiva de asignación

En una asignación: variable := expresión
1) primero se **evalúa** la expresión de derecha y se obtiene un valor,
2) luego se **modifica el valor** de la variable, **perdiéndose el valor anterior.**

```
PROGRAM Asigna3;
VAR a: REAL;
BEGIN
  a:= 1;
  a:= a + a;
  a:= a + a;
  a:= a + a;
  a:= a + a;
END.
```

Traza de los valores almacenados :

a
?
1
2
4
8
16

Primitiva de asignación

En una asignación: variable := expresión
1) primero se evalúa la expresión de derecha y se obtiene un valor,
2) luego se modifica el valor de la variable, perdiéndose el valor anterior.

Un dato sin valor a la derecha de ":=" es un ERROR de programación

```
PROGRAM AsignaMAL;
CONST c = 2.3;
VAR a, b: REAL;
BEGIN
  a:= b;
END.
```

MAL

b no tiene valor

Traza de los valores almacenados en memoria

a	b
?	?
?	?

Observe que el programa `AsignaMAL` es sintácticamente valido y aún así tiene un error.

Intercambiar los valores de las variables

En una asignación: variable := expresión
1) primero se evalúa la expresión de derecha y se obtiene un valor,
2) luego se modifica el valor de la variable, perdiéndose el valor anterior.

Problema: Intercambiar los valores de dos variables `a` y `b` de forma tal que el valor de `a` quede en `b` y el de `b` quede en `a`.

MAL

```
PROGRAM IntercambiaMAL;
VAR a, b: integer;
  BEGIN
  a:= 1;
  b:= 5;
  a:= b;
  b:= a;
END.
```

Observe que `IntercambiaMAL` es sintácticamente valido y aún así tiene un error.

Traza de valores en memoria

a	b
?	?
1	?
1	5
5	5
5	5

Intercambiar los valores de las variables

En una asignación: variable := expresión
1) primero se evalúa la expresión de derecha y se obtiene un valor,
2) luego se modifica el valor de variable, perdiéndose el anterior.

Problema: Intercambiar los valores de dos variables `a` y `b` de forma tal que el valor de `a` quede en `b` y el de `b` quede en `a`.

BIEN

```
PROGRAM IntercambiaBIEN;
VAR a, b, aux: integer;
  BEGIN
  a:= 1;
  b:= 5;
  aux:= a;
  a:= b;
  b:= aux;
END.
```

Preservo el valor de `a` en `aux`.

Traza de valores en memoria

a	b	aux
?	?	?
1	?	?
1	5	?
1	5	1
5	5	1
1	5	1

Primitiva de asignación

El tipo del resultado de la expresión tiene que ser compatible con el tipo de la variable que se quiere modificar (esto se explicará en mayor detalle muy pronto).

```
PROGRAM AsignaMAL;
VAR n,p: INTEGER;
BEGIN
  n:= 5;
  p:= n/2;
END.
```

MAL

n/2 da un resultado REAL y p es de tipo INTEGER

Traza de valores en memoria

n	p
?	?
5	?
5	

Primitiva de asignación

El tipo del resultado de la expresión tiene que ser compatible con el tipo de la variable que se quiere modificar (esto se explicará en mayor detalle muy pronto)**.**

```
PROGRAM AsignaBien;
VAR n,p: INTEGER;
BEGIN
  n:= 5;
  p:= trunc(n/2);
END.
```

BIEN

Traza de valores en memoria

n	p
?	?
5	?
5	2

n/2 da un resultado REAL y p es de tipo INTEGER

Conceptos: programa – código fuente

Un **programa** de computadoras *(computer program),* es un conjunto organizado de instrucciones que están escritas en un lenguaje de programación, y al ser ejecutadas por una computadora resuelven una tarea específica.

Cada lenguaje de programación nos brinda una manera de organizar las instrucciones de un programa.

Pascal es un lenguaje de programación secuencial e imperativo, pensado para crear programas que serán ejecutados secuencialmente en un solo procesador.

El texto de un programa de computadoras se conoce como **código fuente** *(source code).*

Comentarios en el código fuente

En cualquier lugar de un programa en Pascal, se pueden incluir **comentarios** encerrados entre llaves `{}`. También se puede comentar al final de una línea usando dos barras `//`. Estos comentarios serán ignorados en la ejecución el programa pero son muy útiles para los humanos que trabajan con ese código fuente.

```
PROGRAM Transforma;
{Este programa transforma un valor de temperatura
de la escala Celsius a la escala Fahrenheit.}
VAR cel,fah:REAL; //variables para las temperaturas
BEGIN
write('Ingrese temperatura en Celsius ');
readln(cel);
fah:= cel*9/5 + 32; {aquí realiza transformación}
writeln('En Fahrenheit es: ',fah:10:2);
readln; //  Espera a que el usuario presione ENTER
END.
```


Conceptos: tipos de datos

Tipo de Dato: define el conjunto de valores posibles que puede tomar una variable, y las operaciones que pueden aplicarse.

En Pascal (y otros lenguajes de programación) las variables deben ser declaradas indicando su TIPO de DATO.

En Pascal los tipos de datos pueden ser:

1. **predefinidos** (ya tienen un conjunto de valores establecido y proveen operaciones para su uso). Ejemplos que veremos hoy:
   - **INTEGER (enteros)**
   - **REAL (reales)**
   - **CHAR (caracteres)**
   - **BOOLEAN (lógico: verdadero o falso)**
2. **definidos por el programador** (el programador define los valores y las operaciones)

Tipo de dato predefinido de Pascal

Nombre: INTEGER (entero)

Valores: cualquier número entero entre un mínimo y un máximo definido por el compilador usado.

Operaciones predefinidas: que se pueden usar con tipo INTEGER

+ (adición) – (substracción) * (multiplicación)
div (división entera) y **mod** (resto de la división entera).
Operadores relacionales para comparar enteros:
=, >, <, <> (distinto), >= (mayor o igual) y <= (menor o igual)
Obs: los operadores tienen la precedencia usual y los paréntesis () permiten cambiar el orden de evaluación.

Constante predefinida: **MAXINT** (máximo valor INTEGER).

Función predefinida: **SQR** (devuelve el cuadrado (square) de un entero). Ejemplo: SQR(3) = 9. SQR(SQR(3)) =81

Realice una traza y luego pase a la máquina

```
PROGRAM EjemploInteger; {Algunos ejemplos para el tipo entero}
VAR  N1,N2,N3,N4,N5:INTEGER;
       form: INTEGER; {para el "formateo" en pantalla}
BEGIN
   writeln('Máximo entero: ', MAXINT);
   N1 := 2000 mod 2;
   N2 := 2000 div 2 ;
   N3 := SQR(SQR(3));
   N4 := MAXINT;
   form:= 15; //valor para el formateo
   writeln(n1:form, n2:form, n3:form, n4:form);
   N5 := 1+ N4 ;  //¿qué valor toma N5?
   writeln(N5);  //¿por qué será este valor?
END.
```

Tipo de dato predefinido de Pascal

Nombre: **BOOLEAN** (Booleano o lógico)
Valores: (solamente dos) **true**, **false**.
Obs: en Free Pascal (y en Lazarus) son identificadores reservados

Operadores predefinidos:
and (conjunción "y"),
or (disyunción "o")
not (negación "no")
Operadores relacionales: =, >, <, <>, >=, <=

Observe que los operadores relacionales retornan un valor de tipo BOOLEAN (true o false).
Por ejemplo: 5 > 3 retorna true, y 5 <> 5 retorna false

George Boole (1815-1864)



Matemático y filósofo Inglés.
Inventor del "álgebra de Boole", que se transformó en la base de la aritmética computacional moderna.
Es considerado uno de los fundadores del campo de las Ciencias de la Computación.

En 1854 publicó "*An Investigation of the Laws of Thought* " donde desarrolló un sistema de reglas que permite expresar, manipular y simplificar, problemas lógicos y filosóficos cuyos argumentos admiten dos estados (verdadero o falso).

Gracias a su álgebra, hoy en día, podemos manipular operaciones lógicas que son la base de las computadoras y de la informática.

Conceptos: Expresiones Lógicas

Hay sólo dos valores lógicos (también llamados valores de verdad) :
- **verdadero** (`true`)
- **falso** (`false`)

- Los operadores relacionales retornan un valor de verdad: `5 > 3` retorna `true`, `5 <> 5` retorna `false`
- Los operadores lógicos: `and`, `or` y `not`, reciben valores de verdad y retornan valores de verdad.
- Los operadores lógicos permiten construir **expresiones lógicas** que retornarán un valor de verdad.
  Ejemplo: considere `num` de tipo entero:

```
(num > 0) and (num < 10) or not (num = 5)
```

A diferencia de los operadores numéricos, para definir el resultado de un operador lógico alcanza una tabla (llamada Tabla de Verdad)

Tablas de verdad

Una tabla de verdad para un operador lógico, muestra explícitamente el resultado para cada valor posible.

Sea A una expresión lógica cualquiera (esto es, su resultado es verdadero o falso), la tabla de verdad de la negación es la siguiente:

`not A` ← En Pascal

A	no A
true	false
false	true

Ejemplo: A podría representar "tengo señal de wi-fi"

En este caso, si "tengo señal de wi-fi" es verdadero, entonces "**no** tengo señal de wi-fi" es falso.

Tablas de verdad

Sean A y B dos expresiones lógicas cualquiera, las tablas de verdad de la conjunción (y) y de la disyunción (o) son:

		`A and B`	`A or B`
A	B	A y B	A o B
true	true	true	true
true	false	false	true
false	true	false	true
false	false	false	false

Ejemplos:
- podría representar las condiciones para realizar una llamada con la expresión **"tengo señal y tengo saldo"**
- podría representar las condiciones para utilizar Internet con la expresión **"hay red wi-fi o hay red de datos móviles"**

Precedencia de los operadores lógicos

- Los operadores pueden combinarse.
- La precedencia es: **no**, **y**, **o** **(not, and, or)**
- Los paréntesis cambian el orden de evaluación

Por ejemplo, quiero representar una condición con la cual puedo visitar una página de internet.

Compare estas dos expresiones para diferentes casos de prueba:

hay wi-fi o hay red-datos y no (batería = 0)
(hay wi-fi o hay red-datos) y no (batería = 0)

Por ejemplo caso de prueba:
hay wifi = verdadero, **hay red-datos** = falso, **batería =** 0

Compare: no A y B con no (A y B)
Pruebe con: **A** = falso **B**= falso; y luego con **A**= verdadero **B**=falso

Precedencia de los operadores en Pascal

Tabla 12.1: Precedencia de operadores en Free Pascal
http://www.freepascal.org/docs-html/ref/refch12.html

Operador	Precedencia	Categoría
not	La más alta (primero)	Unario
*** / div mod and**	segundo	binario
+ - or	tercero	binario
= <> < > <= >=	La más baja (último)	relacional

- **Para alterar el orden de precedencia en la evaluación se utilizan los () paréntesis.**

Expresiones Lógicas Equivalentes

Dos expresiones lógicas son equivalentes si para todos los casos donde una es verdadera la otra también es verdadera.
Ejemplos:

A > 0 es equivalente a **not (A <= 0)**
not (A or B) no es equivalente a **(not A or not B)**

Importante:
- con un ejemplo alcanza para mostrar que no es equivalente,
- sin embargo, para mostrar que sí es equivalente hay que mostrar para todos los casos posibles.

Expresiones lógicas equivalentes

¿Por qué son importantes las expresiones equivalentes?

Porque la misma condición puede escribirse de diferentes maneras, y hay que buscar la más adecuada

Ejemplo: La condición ***"El número entero N tiene un solo dígito"*** puede representarse con cualquiera de estas cuatro expresiones equivalentes:

(N=1) or (N=2) or (N=3) or (N=4) or (N=5) or (N=6) or (N=7) or (N=8) or (N=9) or (N=0) or (N=-1) or (N=-2) or (N=-3) or (N=-4) or (N=-5) or (N=-6) or (N=-7) or (N=-8) or (N=-9)

(N > -10) and (N < 10)

(N >= -9) and (N <= 9)

N div 10 = 0

¿Cuál usaría?

Realice una traza y luego pase a la máquina

```
PROGRAM EjemploBool; {Algunos ejemplos con el tipo Boolean}
VAR   R1: REAL;
      es_par, es_positivo, mayor_a_maxint: BOOLEAN;
      condicion:BOOLEAN;
BEGIN
Write('ingrese un número'); read(R1);
Es_par :=  (TRUNC(R1) mod 2) <> 1;
es_positivo := R1 >= 0;
Mayor_a_maxint := R1 > MAXINT;
Condicion:= es_par and es_positivo and not mayor_a_maxint;
writeln('Resultado de la expresión lógica:  ', condicion);
readln;
END.
```


Expresiones numéricas y lógicas

- **Al introducir la primitiva de asignación, se mostró que el lado derecho al símbolo ":=" es una expresión que da un valor.**
- **Las expresiones indican ("expresan") como calcular adecuadamente un valor.**
- **Saber construir correctamente expresiones es muy importante porque:**
  - **se utilizan de muchas maneras en un algoritmo (no solo en asignaciones)**
  - **hay expresiones de muchos tipos de valores (no solo numéricos)**

Operadores y valores

- **Operadores numéricos: (ej, + - / * mod div)**
  Toman números y tienen un número por resultado
- **Operadores relacionales: (ej. = > < <> >= <=) Relacionan dos datos del mismo tipo y tienen un resultado que es verdadero o falso.**
- **Operadores lógicos: (ej. and or not)**
  Toman valores del conjunto { verdadero, falso } y su resultado es un valor verdadero o falso.

Expresiones numéricas y lógicas

Tarea, escriba expresiones para:

- **Un número N es mayor a 10**
- **N es mayor a 10 y menor a 100.**
- **N tiene a lo sumo 4 dígitos.**
- **N tiene 4 dígitos (exactamente).**
- **N tiene dos o cuatro dígitos.**
- **N es un número impar.**
- **N es divisible por 7 y divisible por 11 y tiene dos dígitos.**

Hay más ejercitación vea el práctico ☺

El código ASCII

American Standard Code for Information Interchange (ASCII)

Está formado por 256 símbolos, aquí se muestran algunos:

				32		33	!	34	"	35	#	36	$	37	%	38	&	39	'
40	(	41	)	42	*	43	+	44	,	45	-	46	.	47	/	48	0	49	1
50	2	51	3	52	4	53	5	54	6	55	7	56	8	57	9	58	:	59	;
60	<	61	=	62	>	63	?	64	@	65	A	66	B	67	C	68	D	69	E
70	F	71	G	72	H	73	I	74	J	75	K	76	L	77	M	78	N	79	O
80	P	81	Q	82	R	83	S	84	T	85	U	86	V	87	W	88	X	89	Y
90	Z	91	[	92	\	93	]	94	^	95	_	96	`	97	a	98	b	99	c
100	d	101	e	102	f	103	g	104	h	105	i	106	j	107	k	108	l	109	m
110	n	111	o	112	p	113	q	114	r	115	s	116	t	117	u	118	v	119	w
120	x	121	y	122	z	123	{	124	|	125	}	126	~	127		128	Ç	129	ü
130	é	160	á	161	í	162	ó	163	ú	164	ñ	165	Ñ			168	¿		

Tipo de dato predefinido de Pascal

Nombre: CHAR (caracter)

Valores: es el conjunto de los 256 caracteres del código **ASCII** (American Standard Code for Information Interchange)

Operaciones predefinidas: (relacionales) =, >, <, <>, >=, <=

Funciones predefinidas: (ver a continuación)

¿Cómo se diferencia en Pascal entre una variable cuyo identificador es A y el símbolo ASCII A ?

- Para indicar un **valor** de tipo CHAR, se utilizan las comillas simples. Ej : ´a´, ´? ´, ´ + ´, ´ ´, etc.
- En Pascal: ´**A**´ es un valor del tipo char, pero en cambio **A** es un identificador creado por un programador.

Funciones predefinidas del tipo CHAR

CHR: permite obtener un caracter cualquiera a partir de su código ASCII. Ej: chr(65) = ´A´; chr(33) = ´!´.

ORD: dado un caracter cualquiera, devuelve su código ASCII. Ejemplos: ord(´A´) = 65, ord(´!´) = 33.

SUCC: retorna el siguiente ASCII si es que existe. Ejemplos: succ(´A´) = ´B´ , succ(´0´) = ´1´

PRED: retorna el anterior ASCII si es que existe. Ejemplos: pred(´B´) = ´A´, pred(´A´) = ´@´

```
PROGRAM Espaniol;
BEGIN
write(chr(160),chr(130),chr(161),chr(162),chr(163));
write(chr(164),chr(165),chr(168));
END.
```


Tipo de dato predefinido de Pascal

Nombre: REAL (real)

Valores: Se pueden expresar con punto decimal (3.5459), o en notación científica: Por ej. $3.5*10^{-3} = 0.0035$ en Pascal es 3.5E-3 y $1.28*10^{8} = 128000000$ es 1.28E8

Es un subconjunto de los números reales en dos sentidos: (1) tiene mínimo y máximo, y (2) tiene una "precisión" máxima. No se cumple que "entre dos números reales existe siempre otro número real".

Operaciones predefinidas: +, –, ***, /** (división)

operadores relacionales: =, >, <, <>, >=, <=

Funciones predefinidas: ver a continuación

Algunas funciones predefinidas para REAL

Funciones trigonométricas: **SIN, COS y TAN**. Dado un valor de un ángulo (en radianes), devuelven su seno, coseno o tangente. Ejemplos: SIN(0) = 0, COS(0) = 1

Función raíz cuadrada (square root) **SQRT**
Ejemplo: SQRT(4) = 2.0

Función de redondeo ROUND: dado un valor real, devuelve el entero más cercano.
Ejemplos: ROUND(2.9) = 3 ROUND(2.3) = 2

Función truncado TRUNC: dado un valor real, devuelve el entero que resulta de eliminar la parte decimal.
Ejemplos: TRUNC (2.9) = 2 TRUNC(2.3) = 2

Funciones predefinidas para reales/enteros en Pascal

Función	Descripción	Recibe	Retorna
abs	Valor absoluto	real o integer	El mismo tipo recibido
arctan	arctan en radianes	real o integer	real
cos	cosine de un radián	real o integer	real
exp	***e*** a una potencia	real o integer	real
ln	Algoritmo natural	real o integer	real
round	redondeo	real	integer
sin	Seno de un radián	real o integer	real
sqr	Cuandrado (square)	real o integer	El mismo tipo recibido
sqrt	square root (raíz cuad.)	real o integer	real
trunc	truncado	real o integer	integer

http://wiki.freepascal.org/Standard_Functions

Realice una traza y luego pase a la máquina

```
PROGRAM EjemploReal; {Algunos ejemplos con el tipo real}
VAR  N1,N2,N3:INTEGER;
       R1,R2: REAL;
BEGIN
    R1 :=  20 mod 2; // Todo entero es un real
    R2 := MAXINT + 1; // Los reales tienen un rango más amplio
    N1 := TRUNC(2.5);
    N2 := ROUND(2.5);
    N3 := ROUND(3.5); // ¿Por qué redondea así?
    write(R1:5:2, R2:25:2,N1:5,N2:5,N3:5);
    readln; //mantiene consola abierta
END.
```


Conceptos: tipos ordinales en Pascal

De los 4 tipos simples predefinidos que hemos visto, INTEGER,CHAR y BOOLEAN son tipos ordinales (REAL no es ordinal).

Los tipos ordinales tienen estas características:

- **Tienen un primer y último elemento.**
- **Tienen un orden, y para cada elemento está definido el siguiente** (a excepción del último) **y el anterior** (a excepción del primero).

Esto permite utilizar sobre los tipos ordinales las operaciones predefinidas:

- **succ() retorna el siguiente (excepto del último)**
- **pred() retorna el anterior (excepto del primero)**
- **ord() retorna el número de orden**

Continuará